Diferenciální rovnice druhého řádu

Robert Mařík

27. listopadu 2010

1 Homogenní rovnice

Řešením homogenní rovnice může být lineární kombinace, exponenciálních funkcí (reálné různé kořeny charakteristické rovnice) , součin lineárního polynomu a exponenciální funkce (reálný dvojnásobný kořen charakteristické rovnice) anebo součin exponenciální funkce s lineární kombinací goniometrických funkcí (komplexně sdružené kořeny charakteristické rovnice).

Sage code
```
y=function('y',x)
```

Sestavíme rovnici a vyřešíme. Pro kontrolní tisk rovnice na obrazovku použijeme příkaz `show`, protože jinak by se zobrazil pouze výstup posledního řádku. První derivace je v prostředí Sage Notebook zapsána jako `D[0](y)(x)`, druhá derivace je `D[0,0](y)(x)`.

Sage code
```
p,q = -4,4
rovnice = diff(y,x,2)+p*diff(y,x)+q*y == 0
show(rovnice)
desolve(rovnice, y)
```

$$4\,y\left(x\right) - 4\,D[0]\left(y\right)\left(x\right) + D[0, 0]\left(y\right)\left(x\right) = 0$$

$(k_2 x + k_1)e^{(2\,x)}$

Sage code
```
p,q = 8,1
show(diff(y,x,2)+p*diff(y,x)+q*y == 0)
desolve(diff(y,x,2)+p*diff(y,x)+q*y == 0, y)
```

$$y\left(x\right) + 8\,D[0]\left(y\right)\left(x\right) + D[0, 0]\left(y\right)\left(x\right) = 0$$

$k_1 e^{((\sqrt{15}-4)x)} + k_2 e^{(-(\sqrt{15}+4)x)}$

Sage code
```
p,q = 2,2
rovnice = diff(y,x,2)+p*diff(y,x)+q*y == 0
show(rovnice)
desolve(rovnice, y)
```

$$2\,y\left(x\right) + 2\,D[0]\left(y\right)\left(x\right) + D[0, 0]\left(y\right)\left(x\right) = 0$$

$(k_1 \sin\left(x\right) + k_2 \cos\left(x\right))e^{(-x)}$

[0]Podporováno grantem FRVŠ 131/2010.

Sage code
```
p,q = 0,2
rovnice = diff(y,x,2)+p*diff(y,x)+q*y == 0
show(rovnice)
desolve(rovnice, y)
```

$$2\,y\left(x\right) + D[0, 0]\left(y\right)\left(x\right) = 0$$

$k_1 \sin\left(\sqrt{2}x\right) + k_2 \cos\left(\sqrt{2}x\right)$

2 Nehomogenní rovnice

Pokud řešíme nehomogenní rovnici, je řešením součet partikulárního řešení a obecného řešení asociované homogenní rovnice.

Sage code
```
p, q, f = 0, 2, exp(-x)*x
show(diff(y,x,2)+p*diff(y,x)+q*y == f)
desolve(diff(y,x,2)+p*diff(y,x)+q*y == f, y)
```

$$2\,y\left(x\right) + D[0, 0]\left(y\right)\left(x\right) = xe^{\left(-x\right)}$$

$\frac{1}{9}\,(3\,x + 2)e^{\left(-x\right)} + k_1 \sin\left(\sqrt{2}x\right) + k_2 \cos\left(\sqrt{2}x\right)$

3 Počáteční podmínky

Pro zadání počátečních podmínek (u homogenních i nehomogenních rovnic je to stejné) použijeme volitelný parametr příkazu `desolve`, parametr `ics`.

Sage code
```
p, q, f = 0, 2, exp(-x)*x
desolve(diff(y,x,2)+p*diff(y,x)+q*y == f, y, ics=[0,2,-2])
```

$\frac{1}{9}\,(3\,x + 2)e^{\left(-x\right)} - \frac{19}{18}\,\sqrt{2}\sin\left(\sqrt{2}x\right) + \frac{16}{9}\,\cos\left(\sqrt{2}x\right)$

Pro kontrolu si nejprve označíme výsledek jako funkci g a potom dosadíme počáteční podmínku $x = 0$ do funkce $g(x)$ a derivace $g'(x)$.

Sage code
```
g(x)=_
g(0),diff(g)(x=0)
```

$(2, -2)$

Jiná možnost jak započítat počáteční podmínky je použít stejnou cestu, jako bychom počítali ručně: dosadíme do y a y', řešením soustavy rovnic určíme konstanty k_1 a k_2 a tyto konstanty použijeme v obecném řešení.

Sage code
```
k1,k2 = var('k1 k2')
res(x)=desolve(diff(y,x,2)+p*diff(y,x)+q*y == f, y)
soustava = [ res(x=0) == 2, diff(res(x),x).subs(x=0) == -2 ]
koeficienty=solve( soustava, [k1,k2], solution_dict=True)
res(x).subs(koeficienty[0])
```

$\frac{1}{9}\,(3\,x + 2)e^{\left(-x\right)} - \frac{19}{18}\,\sqrt{2}\sin\left(\sqrt{2}x\right) + \frac{16}{9}\,\cos\left(\sqrt{2}x\right)$

Nakreslíme graf patrikulárního řešení z předchozího příkladu. Díky klesající exponenciální funkci e^{-x} se partikulární řešení blíží velmi rychle k nule a zůstane lineární kombinace sinusoidy a kosinusoidy z řešení asociované homogenní rovnice (červeně).

Sage code
```
res_hom(x)=(desolve(diff(y,x,2)+p*diff(y,x)+q*y == 0, y)).subs(koeficienty[0])
show(plot(res(x).subs(koeficienty[0]),(x,0,10))\
+plot(res_hom(x),(x,0,10), rgbcolor='red', zorder=-1))
```



4 Metoda neurčitých koeficientů

Sage umí řešit diferenciální rovnice plně automaticky. Nicméně, můžeme jej použít i když chceme simulovat postup při ručním počítání. Pokusme se vyřešit rovnici $y'' + 3y' + 2y = x^2e^{-2x}$

Sage code
```
p,q,f=3,2,x^2*exp(-2*x)
a,b,c,x = var('a,b,c,x')
y = x*(a*x^2+b*x+c)*exp(-2*x) # tvar partikularniho reseni
y
```

$(ax^2 + bx + c)xe^{(-2\,x)}$

Sage code
```
diff(y,x,2)+p*diff(y,x)+q*y    # dosazeni do leve strany rovnice
```

$-(2\,ax + b)xe^{(-2\,x)} + 2\,axe^{(-2\,x)} + 2\,(2\,ax + b)e^{(-2\,x)} - \left(ax^2 + bx + c\right)e^{(-2\,x)}$

Sage code
```
rovnice = _ == f # pripojeni prave strany a sestaveni rovnice
rovnice
```

$-(2\,ax + b)xe^{(-2\,x)} + 2\,axe^{(-2\,x)} + 2\,(2\,ax + b)e^{(-2\,x)} - \left(ax^2 + bx + c\right)e^{(-2\,x)} = x^2e^{(-2\,x)}$

Sage code
```
rovnice/exp(-2*x)  # vydeleni exponencialnim clenem
```

$-\left((2\,ax + b)xe^{(-2\,x)} - 2\,axe^{(-2\,x)} - 2\,(2\,ax + b)e^{(-2\,x)} + \left(ax^2 + bx + c\right)e^{(-2\,x)}\right)e^{(2\,x)} = x^2$

Sage code
```
r=expand(_)     # roznasobeni zavorek
r
```

$-3\,ax^2 + 6\,ax - 2\,bx + 2\,b - c = x^2$
Sestavení rovnic pro neurčité koeficienty je malinko komplikovanější. Rovnici mezi dvěma polynomy máme v proměnné `r`. Nejprve musíme nalézt levou a pravou stranu pomocí `lhs()` a `rhs()` a poté nalezneme koeficient u mocniny x^i pomocí `coefficient(x,i)`. Koeficient z levé strany položíme roven odpovídajícímu koeficientu ze strany pravé a opakujeme pro $i = 0, 1, 2$.

Sage code
```
soustava = [r.lhs().coefficient(x,i)==r.rhs().coefficient(x,i) for i in range(3)]
soustava
```

$[2\,b - c = 0, 6\,a - 2\,b = 0, -3\,a = 1]$
Soustavu vyřešíme příkazem `solve`, druhým parametrem je n-tice neznámých.

Sage code
```
solve(soustava,[a,b,c])
```

$\left[\left[a = \left(-\frac{1}{3}\right), b = (-1), c = (-2)\right]\right]$
Řešení vidíme, na obrazovce. Aby se nám pohodlně dosazovalo do původního předpisu, vyřešíme soustavu ještě jednou, nyní s parametrem `solution_dict=True`.

Sage code
```
koeficienty = solve(soustava,[a,b,c], solution_dict=True)
koeficienty
```

$\left[\left\{c:\ -2, b:\ -1, a:\ -\frac{1}{3}\right\}\right]$

Sage code
```
y.subs(koeficienty[0]).expand()
```

$-\frac{1}{3}\,x^3e^{(-2\,x)} - x^2e^{(-2\,x)} - 2\,xe^{(-2\,x)}$
Pro srovnání stejná diferenciální rovnice, vyřešená příkazem `desolve`. Dostáváme obecné řešení.

Sage code
```
y = function('y',x)
reseni = desolve(diff(y,x,2)+p*diff(y,x)+q*y == f,y)
expand(reseni)
```

$-\frac{1}{3}\,x^3e^{(-2\,x)} - x^2e^{(-2\,x)} + k_1e^{(-x)} + k_2e^{(-2\,x)} - 2\,xe^{(-2\,x)} - 2\,e^{(-2\,x)}$
Vidíme že partikulární řešení vypočtená oběma metodami se shodují (až na člen $-2e^{-2x}$, který je možno zahrnout do členu k_2e^{-2x} a přejmenovat konstantu "$k_2 - 2$"na K_2).

Sage code
```
k1,k2,K2=var('k1,k2,K2')
expand(reseni.subs(k2=K2+2))
```

$-\frac{1}{3}\,x^3e^{(-2\,x)} - x^2e^{(-2\,x)} + K_2e^{(-2\,x)} + k_1e^{(-x)} - 2\,xe^{(-2\,x)}$

Sage code
```
expand(reseni.subs(k2=K2+2)).subs({K2:0,k1:0})
```

$-\frac{1}{3}\,x^3e^{(-2\,x)} - x^2e^{(-2\,x)} - 2\,xe^{(-2\,x)}$
Pokud podobný postup použijeme (nyní již zhuštěně) pro špatný tvar partikulárního řešení, narazíme na problémy - soustava nemá řešení.

Sage code
```
a,b,c,x = var('a,b,c,x')
p,q,f = 3,2,x^2*exp(-2*x)
y = (a*x^2+b*x+c)*exp(-2*x) # spatny tvar partikularniho reseni
rovnice = diff(y,x,2)+p*diff(y,x)+q*y == f
r=expand(rovnice/exp(-2*x))
soustava = [r.lhs().coefficient(x,i)==r.rhs().coefficient(x,i) for i in range(3)]
soustava
```

$[2a - b = 0, -2a = 0, 0 = 1]$

Sage code

```
solve(soustava,[a,b,c]) # soustava nema reseni
```

$[]$